Travaux Pratiques - Série 3, TP° 1
Résolution d'équations différentielles

L'objectif de ce TP est de montrer comment Scilab permet de résoudre de manière rapprochée toute équation (ou système d'équations) différentielle.

1 Programmation en Scilab : définition d'une fonction externe

Nous avons jusqu'à présent utilisé des fonctions déjà prédéfinies en Scilab. Or le logiciel nous permet de définir nos propres fonctions. Une fois définies, ces fonctions peuvent être appelées et utilisées de la même manière que les fonctions prédéfinies de Scilab.

Prenons un exemple simple : la définition d'une fonction qui calcule l'impédance complexe d'un circuit RLC série :

```
function [Z]=impedance(f,R,L,C)
    if(C>0) then
        Z=R+%i*2*%pi*f*L+1/(%i*2*%pi*f*C)
    else
        Z=R+%i*2*%pi*f*L
    end
endfunction
```

Dans le script précédant :
- `impedance` est le nom de la fonction qu'on définit
- `f, R, L, C` sont les données (ou arguments) d'entrée de la fonction
- `Z` est la variable de sortie de la fonction ; dans cette variable va être stockée la valeur de la fonction.

La syntaxe générale pour la définition d'une fonction est donc la suivante :

```
function <arguments_sortie>=<nom_de_la_fonction>(<arguments_entree>)
  <instructions>
endfunction
```

Exercices.

1. Écrire et tester la fonction `impedance`.
2. Écrire une fonction qui calcule la résistance équivalente de 3 résistances mises en parallèle. Elle prendra en argument d'entrée les 3 résitances R_1, R_2 et R_3 et elle renverra la valeur de la résistance équivalente.
3. Écrire une fonction qui résout une équation du second degré à coefficients réels (de la forme $ax^2 + bx + c$). Elle prendra en argument les coefficients `a, b c` et elle renverra les deux solutions (réeles ou complexes).

   ```
   function [x1,x2]=resoudre2d(a,b,c)
     ...
   endfunction
   ```

   Vérifier votre fonction :

   ```
   [x1,x2]=resoudre2d(1,5,6)
   [y1,y2]=resoudre2d(1,1,1)
   [z1,z2]=resoudre2d(1,2,1)
   ```

Quel résultat est affiché si on écrit `x1=resoudre2d(1,5,6)` à la place de `[x1,x2]=resoudre2d(1,5,6)` ?

4. Scilab contient des fonctions qui permettent de définir un polynôme et trouver ses racines. Ainsi la commande `p = poly([6,5,1],'x',"coeff")` permet de définir le polynôme $p = x^2 + 5x + 6$.
   Attention : les coefficients sont donnés en commençant par celui du degré 0. Par exemple, la commande `p = poly([0,0,6,5,1],'x',"coeff")` permet de définir le polynôme $p = x^4 + 5x^3 + 6x^2$.
   La commande `roots(p)` permet d'obtenir les racines du polynôme précédent.
   Comparer les racines obtenues avec votre fonction à celles obtenues avec la fonction `roots` de Scilab.

2 Systèmes du premier ordre : circuits de base

2.1 Circuit $R-C$: décharge du condensateur

On considére le circuit représenté à la Figure 1. La tension aux bornes du condensateur est initialement égale à $E = 10\,\text{V}$.



Figure 1 – Montage 1

1. Quelle est la relation liant le courant $i(t)$ et la tension $v_c(t)$? A partir de la loi des mailles, exprimer l'équation différentielle du premier ordre pour $v_C(t)$.
2. En TD d'OL2, vous avez résolu analytiquement cette équation différentielle linéaire (du 1er ordre, à coefficients constants) et trouvé l'expression analytique de la tension aux bornes du condensateur :

$$v_{C,theo}(t) = E \times e^{\frac{-t}{\tau}}$$

   où $R \times C = \tau$
   Avec Scilab, on peut résoudre numériquement cette équation différentielle. Le script ci-dessous permet de déterminer la solution numérique, à l'aide de la fonction `ode`.

```
clear; clf;
//  ED Simple  - DECHARGE DU CONDENSATEUR
// dy/dt=-1/tau*y, y(0)=10

function derivee_tension_condo=fd(t, tension_condo)
    derivee_tension_condo=-tension_condo/tau
endfunction
R=1.e+3;C=5.e-3; tau=R*C;
tension_condo_initiale=10;t0=0;
t=0:0.1:26
tension_condo=ode(tension_condo_initiale,t0,t,fd)
```

```
subplot(2,1,1)
plot(t,tension_condo,'-b')
xtitle("Circuit RC: DECHARGE DU CONDENSATEUR","temps (s)", "$v_C{(t)}$")
xgrid()

// TANGENTE THEORIQUE A LA COURBE DE DECHARGE A L'ORIGINE
v_ctan=[tension_condo_initiale;0]
plot([0;tau], v_ctan,'k')
legend("$v_C$","tangente ")
// VALEURS CARACTERISTIQUES de vC A: t=tau,3*tau et 5*tau
m=[0; tension_condo_initiale]
plot([tau;tau], m,'r')
plot([3*tau;3*tau], m,'r')
plot([5*tau;5*tau], m,'r')
// DIFFERENCE ENTRE LA SOLUTION THEORIQUE ET CELLE NUMERIQUE
subplot(2,1,2)
vc_theo=tension_condo_initiale*exp(-t/tau)
plot(t,abs(tension_condo-vc_theo),'-om')
xtitle("DIFFERENCE ENTRE LA SOLUTION THEORIQUE ET CELLE NUMERIQUE",...
        "temps (s)","$|v_C^{num}-v_C^{theo}|{(t)}$")
```

3. Copier ce script et étudier le.
4. Sur une autre figure tracer l'évolution du courant $i(t)$.

A RETENIR

Pour obtenir la solution numérique d'une équation différentielle du **premier ordre** avec Scilab, il faut écrire une instruction de type :

y=ode(y0,t0,t,f)

où y0 est la valeur initiale à t0 de la fonction inconnue y et t est un vecteur d'abscisses (temporelles dans notre cas) pour lesquelles la solution numérique sera déterminée. Enfin f est une fonction correspondant à :

ẏ=f(t,y)

Il s'agit de l'équation différentielle étudiée.
Cette démarche s'applique aussi bien aux équations différentielles linéaires que non-linéaires ou encore à coefficients constants ou non-constants.
Pour résoudre des équations différentielles **d'ordre supérieur**, il faudra les transformer en plusieurs équations du **premier ordre** en introduisant des variables correspondant aux dérivées intermédiaires.

2.2 Circuit $R-C$: charge du condensateur

On considère le circuit représenté à la Figure 2.



Figure 2 – Montage 2

1. Quelle est la relation liant le courant $i(t)$ et la tension $v_s(t)$? A partir de la loi des mailles, exprimer l'équation différentielle du premier ordre liant $v_e(t)$ et $v_s(t)$.
2. La tension d'entrée est constante : $v_e(t) = E$, et la tension de sortie est initialement nulle (condensateur déchargé).
   Ecrivez le script permettant de résoudre numériquement cette équation différentielle et tracer l'évolution de la tension de sortie $v_s(t)$ en fonction du temps.
   Application numérique : $R = 5$ kΩ, $C = 500\mu$F et $E = 5$ V.
3. La tension d'entrée est constante : $v_e(t) = E$, et la tension de sortie est initialement égale à $E/2$ (condensateur chargé). Tracer l'évolution de la tension de sortie en fonction du temps sur le même graphique.

2.3 Circuit $L - R$

On considère le circuit représenté à la Figure 3.



Figure 3 – Montage 3

1. Quelle est la relation liant le courant $i(t)$ et la tension $v_s(t)$? Quelle est la relation liant le courant $i(t)$ et la tension aux bornes de l'inductance L ? A partir de la loi des mailles, exprimer l'équation différentielle du premier ordre liant $v_e(t)$ et $v_s(t)$.
2. La tension d'entrée est constante : $v_e(t) = E$, et la tension de sortie est initialement nulle. Ecrivez le script permettant de résoudre numériquement cette équation différentielle et tracer l'évolution de la tension de sortie $v_s(t)$ en fonction du temps.

   Application numérique : $R = 1\Omega$, $L = 1$H et $E = 5$ V.